
README

Gnat
A nats.io client for Elixir.
The goals of the project are resiliency, performance, and ease of use.
Hex documentation available here: https://hex.pm/packages/gnat


  
    
  
  Usage


{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222})
# Or if the server requires TLS you can start a connection with:
# {:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, tls: true})

{:ok, subscription} = Gnat.sub(gnat, self(), "pawnee.*")
:ok = Gnat.pub(gnat, "pawnee.news", "Leslie Knope recalled from city council (Jammed)")
receive do
  {:msg, %{body: body, topic: "pawnee.news", reply_to: nil}} ->
    IO.puts(body)
end

  
    
  
  Authentication


# with user and password
{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, username: "joe", password: "123", auth_required: true})

# with token
{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, token: "secret", auth_required: true})

# with an nkey seed
{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, nkey_seed: "SUAM...", auth_required: true})

# with decentralized user credentials (JWT)
{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, nkey_seed: "SUAM...", jwt: "eyJ0eX...", auth_required: true})

# connect to NGS with JWT
{:ok, gnat} = Gnat.start_link(%{host: "connect.ngs.global", tls: true, jwt: "ey...", nkey_seed: "SUAM..."})

  
    
  
  TLS Connections


NATS Server is often configured to accept or require TLS connections.
In order to connect to these clusters you'll want to pass some extra TLS settings to your Gnat connection.
# using a basic TLS connection
{:ok, gnat} = Gnat.start_link(%{host: "127.0.0.1", port: 4222, tls: true})

# Passing a Client Certificate for verification
{:ok, gnat} = Gnat.start_link(%{tls: true, ssl_opts: [certfile: "client-cert.pem", keyfile: "client-key.pem"]})

  
    
  
  Resiliency


If you would like to stay connected to a cluster of nats servers, you should consider using Gnat.ConnectionSupervisor .
This can be added to your supervision tree in your project and will handle automatically re-connecting to the cluster.
For long-lived subscriptions consider using Gnat.ConsumerSupervisor .
This can also be added to your supervision tree and use a supervised connection to re-establish a subscription.
It also handles details like handling each message in a supervised process so you isolate failures and get OTP logs when an unexpected error occurs.

  
    
  
  Services


If you supply a module that implements the Gnat.Services.Server behavior and the service_definition 
configuration field to a Gnat.ConsumerSupervisor, then this client will automatically take care
of exposing the service to discovery, responding to pings, and maintaining and exposing statistics like request and error counts, and processing times.

  
    
  
  Instrumentation


Gnat uses telemetry to make instrumentation data available to clients.
If you want to record metrics around the number of messages or latency of message publishes, subscribes, requests, etc you can do the following in your project:
iex(1)> metrics_function = fn(event_name, measurements, event_meta, config) ->
  IO.inspect([event_name, measurements, event_meta, config])
  :ok
end
#Function<4.128620087/4 in :erl_eval.expr/5>
iex(2)> names = [[:gnat, :pub], [:gnat, :sub], [:gnat, :message_received], [:gnat, :request], [:gnat, :unsub]]
[
  [:gnat, :pub],
  [:gnat, :sub],
  [:gnat, :message_received],
  [:gnat, :request],
  [:gnat, :unsub],
  [:gnat, :service_request],
  [:gnat, :service_error]
]
iex(3)> :telemetry.attach_many("my listener", names, metrics_function, %{my_config: true})
:ok
iex(4)> {:ok, gnat} = Gnat.start_link()
{:ok, #PID<0.203.0>}
iex(5)> Gnat.sub(gnat, self(), "topic")
[[:gnat, :sub], %{latency: 128000}, %{topic: "topic"}, %{my_config: true}]
{:ok, 1}
iex(6)> Gnat.pub(gnat, "topic", "ohai")
[[:gnat, :pub], %{latency: 117000}, %{topic: "topic"}, %{my_config: true}]
[[:gnat, :message_received], %{count: 1}, %{topic: "topic"}, %{my_config: true}]
:ok
The pub , sub , request , and unsub events all report the latency of those respective calls.
The message_received event reports a number of messages like %{count: 1} because there isn't a good latency metric to report. Any microservices managed by a consumer supervisor will also report service_request and service_error. In addition to the :topic metadata, microservices will also include :endpoint and :group (which can be nil) in their telemetry reports.
All of the events (except unsub ) include metadata with a :topic key so you can split your metrics by topic.

  
    
  
  Benchmarks


Part of the motivation for building this library is to get better performance.
To this end, there is a bench branch on this project which includes a server.exs and client.exs that can be used for benchmarking various scenarios.
As of this commit, the latest benchmark on a 16-core server shows that you can make 170k+ req/sec or up to 192MB/sec.
The bench/*.exs files also contain some straight-line single-CPU performance tests.
As of this commit my 2018 MacBook pro shows.
		ips	average	deviation	median
	parse-128	487.67 K	2.19 μs	±1701.54%	2 μs
	pub - 128	96.37 K	10.38 μs	±102.94%	10 μs
	req-reply-128	8.32 K	120.16 μs	±23.68%	114 μs


  
    
  
  Development


Before running the tests make sure you have a locally running copy of nats-server (installation instructions). By default, tests are run with no authentication. Make sure your NATS configuration contains no users, or has an account with no_auth_user explicitly enabled.
We currently use version 2.10.24 in CI, but anything higher than 2.2.0 should be fine.
Versions from 0.9.6 up to 2.2.0 should work fine for everything except header support.
Make sure to enable jetstream with the nats-server -js argument and you might also want to enable debug and verbose logging if you're trying to understand the messages being sent to/from nats (ie nats-server -js -D -V).
The typical mix test will run all the basic unit tests.
You can also run the multi_server set of tests that test connectivity to different
nats-server configurations. You can run these with mix test --only multi_server .
The tests will tell you how to start the different configurations.
There are also some property-based tests that generate a lot of test cases.
You can tune how many test cases by setting the environment variable N=200 mix test --only property (default it 100).
For more details you can look at how Github runs these things in the CI flow.


  

    
Changelog
    


  
    
  
  1.8


	Integrated the jetstream functionality into this client directly https://github.com/nats-io/nats.ex/pull/146
	Add ability to list KV buckets https://github.com/nats-io/nats.ex/pull/152
	Improve CI Reliability https://github.com/nats-io/nats.ex/pull/154
	Bugfix to treat no streams as an empty list rather than a null https://github.com/nats-io/nats.ex/pull/155
	Added supported for allow_direct and mirror_direct attributes of streams https://github.com/nats-io/nats.ex/pull/161
	Added support for discard_new_per_subject attribute of streams https://github.com/nats-io/nats.ex/pull/163
	Added support for Object.list_buckets https://github.com/nats-io/nats.ex/pull/169


  
    
  
  1.7


	Added support for the NATS services API, letting developers participate in service discovery and stats https://github.com/nats-io/nats.ex/pull/141
	A bugfix to remove the queue_group from a service config and some optimization for the services API https://github.com/nats-io/nats.ex/pull/145


  
    
  
  1.6


	added the no_responders behavior https://github.com/nats-io/nats.ex/pull/137


  
    
  
  1.5


	add the inbox_prefix option https://github.com/nats-io/nats.ex/pull/121
	add the Gnat.server_info/1 function https://github.com/nats-io/nats.ex/pull/124
	fix header parsing issue https://github.com/nats-io/nats.ex/pull/125


  
    
  
  1.4


	add the Gnat.request_multi/4 function https://github.com/nats-io/nats.ex/pull/120
	add elixir 1.13 to the test matrix


  
    
  
  1.3


	adding support for sending and receiving headers https://github.com/nats-io/nats.ex/pull/116


  
    
  
  1.2


	Gnat.Server behaviour with support in the ConsumerSupervisor https://github.com/nats-io/nats.ex/compare/1b1adc85e4b28231218ef87c7fc3445fce854377...b24a7e14325b51fbb93fde7e3d891d18b4fa8afb
	avoid logging sensitive credentials https://github.com/nats-io/nats.ex/pull/105
	deprecate Gnat.ping, improved typespecs https://github.com/nats-io/nats.ex/pull/103 
	relax the version constraint on nimble_parsec https://github.com/nats-io/nats.ex/issues/112


  
    
  
  1.1


	add support for nkeys and NGS https://github.com/nats-io/nats.ex/pull/101
	Fix supervisor ConsumerSuperivsor crash https://github.com/nats-io/nats.ex/pull/96


  
    
  
  1.0


	Make supervisors officially supported https://github.com/nats-io/nats.ex/pull/96


  
    
  
  0.7.0


	update to telemetry 0.4 https://github.com/nats-io/nats.ex/pull/86 and https://github.com/nats-io/nats.ex/pull/87
	support for token authentication https://github.com/nats-io/nats.ex/pull/92
	support elixir 1.9 https://github.com/nats-io/nats.ex/pull/93


  
    
  
  0.6.0


	Dropped support for Erlang < 19 and Elixir <= 1.5
	Added Telemetry to the project (thanks @rubysolo)
	Switched to nimble_parsec for parsing	Updated benchmarking/performance information. We can now do 170k requests per second on a 16-core server.


	Fixed a bug around re-subscribing for the ConsumerSupervisor
	Pass sid when delivering message (thanks @entone)
	Documentation fixes from @deini and @johannestroeger


  
    
  
  0.5.0


	Dropped support for Elixir 1.4 and OTP 18 releases. You will need to use Elixir 1.5+ and OTP 19+.
	Switched to running our tests against gnatsd 1.3.0



  

    
Overview
    

Jetstream is a distributed persistence system
built-in to NATS. It provides a streaming system that lets you capture streams
of events from various sources and persist these into persistent stores, which you can immediately
or later replay for processing.
This library exposes interfaces for publishing, consuming and managing Jetstream services. It builds
on top of Gnat, the officially supported Elixir client for NATS.
	Let's get Jetstream up and running
	Using Broadway with Jetstream
	Pull Consumer API
	Create, update and delete Jetstream streams and consumers via Elixir



  

    
Getting Started
    

In this guide, we're going to learn how to install Jetstream in your project and start consuming
messages from your streams.

  
    
  
  Starting Jetstream


The following Docker Compose file will do the job:
version: "3"
services:
  nats:
    image: nats:latest
    command:
      - -js
    ports:
      - 4222:4222
Save this snippet as docker-compose.yml and run the following command:
docker compose up -d

Let's also create Jetstream stream where we will publish our hello world messages:
nats stream add HELLO --subjects="greetings"

Tip
You can also manage Jetstream streams and consumers via Elixir. You can see more details in
this guide.

  
    
  
  Adding Jetstream and Gnat to an application


To start off with, we'll generate a new Elixir application by running this command:
mix new hello_jetstream --sup
We need to have a supervision tree
up and running in your app, and the --sup option ensures that.
To add Jetstream to this application, you need to add Jetstream
and Gnat libraries to your deps definition in our mix.exs file.
Fill exact version requirements from each package Hex.pm pages.
defp deps do
  [
    {:gnat, ...},
    {:jetstream, ...}
  ]
end
To install these dependencies, we will run this command:
mix deps.get

Now let's connect to our NATS server. To do this, you need to start Gnat.ConnectionSupervisor
under our application's supervision tree. Add following to lib/hello_jetstream/application.ex:
def start(_type, _args) do
  children = [
    ...

    # Create NATS connection
    {Gnat.ConnectionSupervisor,
      %{
        name: :gnat,
        connection_settings: [
          %{host: "localhost", port: 4222}
        ]
      }},
  ]

  ...
This piece of configuration will start Gnat processes that connect to the NATS server and allow
publishing and subscribing to any subjects. Jetstream operates using plain NATS subjects which
follow specific naming and message format conventions.
Let's now create a pull consumer which will subscribe a specific Jetstream stream and print
incoming messages to standard output.

  
    
  
  Creating a pull consumer


Jetstream requires us to allocate a view/cursor of the stream that our consumer will operate on.
In Jetstream terminology, this view is called a consumer (Funnily enough we've just implemented
a consumer in our code, coincidence?). Jetstream
documentation
offers great insights on benefits of having this separate concept so we won't duplicate work here.
Jetstream offers two stream consuming modes: push and pull.
In push mode, Jetstream will simply send messages to selected consumers immediately when they are
received. This approach does offer congestion control, so it is not recommended for high-volume
and/or reliability sensitive streams. You do not really need this library to implement push
consumer because all building blocks are in Gnat library. You can read more about push consumers
in this guide.
On the other hand, in pull mode consumers ask Jetstream for more messages when they are ready
to process them. This is the recommended approach for most use cases and we will proceed with it
in this guide.
This is just a brief outline
For more details about differences between consumer modes, consult
Jetstream documentation.
Let's create a pull consumer module within our application at
lib/hello_jetstream/logger_pull_consumer.ex:
defmodule HelloJetstream.LoggerPullConsumer do
  use Jetstream.PullConsumer

  def start_link([]) do
    Jetstream.PullConsumer.start_link(__MODULE__, [])
  end

  @impl true
  def init([]) do
    {:ok, nil, connection_name: :gnat, stream_name: "HELLO", consumer_name: "LOGGER"}
  end

  @impl true
  def handle_message(message, state) do
    IO.inspect(message)
    {:ack, state}
  end
end
Pull Consumer is a regular GenServer and it takes a reference to Gnat.ConnectionSupervisor
along with names of Jetstream stream and consumer as options passed to
Jetstream.PullConsumer.start* functions. These options are passed as keyword list in third element
of tuple returned from the c:Jetstream.PullConsumer.init/1 callback.
The only required callbacks are well known gen server's c:Jetstream.PullConsumer.init/1 and
c:Jetstream.PullConsumer.handle_message/2, which takes new message as its first argument and
is expected to return an ACK action instructing underlying process loop what to do with this
message. Here we are asking it to automatically send for us an ACK message back to Jetstream.
Let's now create a consumer in our NATS server. We will call it LOGGER as we plan to let it simply
log everything published to the stream.
nats consumer add --pull --deliver=all HELLO LOGGER

Now, let's start our pull consumer under application's supervision tree.
def start(_type, _args) do
  children = [
    ...

    # Jetstream Pull Consumer
    HelloJetstream.LoggerPullConsumer,
  ]

  ...
Let's now publish some messages to our HELLO stream, so something will be waiting for our
application to be read when it starts.

  
    
  
  Publishing messages to streams


Jetstream listens on regular NATS subjects, so publishing messages is dead simple with Gnat.pub/3:
Gnat.pub(:gnat, "greetings", "Hello World")
Or via NATS CLI:
nats pub greetings "Hello World"

That's it! When you run your app, you should see your messages being read by your application.


  

    
Managing Streams and Consumers
    

Jetstream provides a JSON API for managing streams and consumers.
This library exposes this API via interactions with the Jetstream.Api.Stream and Jetstream.Api.Consumer modules.
These modules act as native wrappers for the API and do not attempt to simplify any of the common use-cases.
As this library matures we may introduce a separate layer of functions to handle these scenarios, but for now our aim is to provide full access to the Jetstream API.


  

    
Push based consumer
    

# Start a nats server with jetstream enabled and default configs
# Now run the following snippets in an IEx terminal
alias Jetstream.API.{Consumer,Stream}

# Setup a connection to the nats server and create the stream/consumer
# This is the equivalent of these two nats cli commands
#   nats stream add TEST --subjects="greetings" --max-msgs=-1 --max-msg-size=-1 --max-bytes=-1 --max-age=-1 --storage=file --retention=limits --discard=old
#   nats consumer add TEST TEST --target consumer.greetings --replay instant --deliver=all --ack all --wait=5s --filter="" --max-deliver=10
{:ok, connection} = Gnat.start_link()
stream = %Stream{name: "TEST", subjects: ["greetings"]}
{:ok, _response} = Stream.create(connection, stream)
consumer = %Consumer{stream_name: "TEST", name: "TEST", deliver_subject: "consumer.greetings", ack_wait: 5_000_000_000, max_deliver: 10}
{:ok, _response} = Consumer.create(connection, consumer)

# Setup Consuming Function
defmodule Subscriber do
  def handle(msg) do
    IO.inspect(msg)
    case msg.body do
      "hola" -> Jetstream.ack(msg)
      "bom dia" -> Jetstream.nack(msg)
      _ -> nil
    end
  end
end

# normally you would add the `ConnectionSupervisor` and `ConsumerSupervisor` to your supervisrion tree
# here we start them up manually in an IEx session
{:ok, _pid} = Gnat.ConnectionSupervisor.start_link(%{
  name: :gnat,
  backoff_period: 4_000,
  connection_settings: [
    %{}
  ]
})
{:ok, _pid} = Gnat.ConsumerSupervisor.start_link(%{
  connection_name: :gnat,
  consuming_function: {Subscriber, :handle},
  subscription_topics: [
    %{topic: "consumer.greetings"}
  ]
})

# now publish some messages into the stream
Gnat.pub(:gnat, "greetings", "hello") # no ack will be sent back, so you'll see this message received 10 times with a 5sec pause between each one
Gnat.pub(:gnat, "greetings", "hola") # an ack is sent back so this will only be received once
Gnat.pub(:gnat, "greetings", "bom dia") # a -NAK is sent back so you'll see this received 10 times very quickly


  

    
Using Broadway with Jetstream
    

Broadway is a library which allows building concurrent and multi-stage data ingestion and data
processing pipelines with Elixir easily. You can learn about it more in
Broadway documentation.
Jetstream library comes with tools necessary to use NATS Jetstream with Broadway.

  
    
  
  Getting started


In order to use Broadway with NATS Jetstream you need to:
	Setup a NATS Server with JetStream turned on
	Create stream and consumer on NATS server
	Configure Gnat connection in your Elixir project
	Configure your project to use Broadway

In this guide, we are going to focus on the fourth point. To learn how to start Jetstream locally
with Docker Compose and then add Gnat and Jetstream to your application, see the Starting Jetstream
section in Getting Started guide.

  
    
  
  Adding Broadway to your application


Once we have NATS with JetStream running and the stream and consumer we are going to use are
created, we can proceed to adding Broadway to our project. First, put :broadway to the list of
dependencies in mix.exs.
defp deps do
  [
    ...
    {:broadway, ...version...},
    ...
  ]
end
Visit Broadway page on Hex.pm to check for current version
to put in deps.
To install the dependencies, run:
mix deps.get


  
    
  
  Defining the pipeline configuration


The next step is to define your Broadway module. We need to implement three functions in order
to define a Broadway pipeline: start_link/1, handle_message/3 and handle_batch/4.
Let's create start_link/1 first:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(
      __MODULE__,
      name: MyBroadwayExample,
      producer: [
        module: {
          OffBroadway.Jetstream.Producer,
          connection_name: :gnat,
          stream_name: "TEST_STREAM",
          consumer_name: "TEST_CONSUMER"
        },
        concurrency: 10
      ],
      processors: [
        default: [concurrency: 10]
      ],
      batchers: [
        default: [
          concurrency: 5,
          batch_size: 10,
          batch_timeout: 2_000
        ]
      ]
      ...
    )
  end

  ...callbacks..
end
All start_link/1 does is just delegating to Broadway.start_link/2.
To understand what all these options mean and to learn about other possible settings, visit
Broadway documentation.
The part that interests us the most in this guide is the producer.module. Here we're choosing
OffBroadway.Jetstream.Producer as the producer module and passing the connection options,
such as Gnat process name and stream name. For full list of available options, visit
Producer documentation.

  
    
  
  Implementing Broadway callbacks


Broadway requires some callbacks to be implemented in order to process messages. For full list
of available callbacks visit
Broadway documentation.
A simple example:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  ...start_link...

  def handle_message(_processor_name, message, _context) do
    message
    |> Message.update_data(&process_data/1)
    |> case do
      "FOO" -> Message.configure_ack(on_success: :term)
      "BAR" -> Message.configure_ack(on_success: :nack)
      message -> message
    end
  end

  defp process_data(data) do
    String.upcase(data)
  end

  def handle_batch(_, messages, _, _) do
    list = messages |> Enum.map(fn e -> e.data end)
    IO.puts("Got a batch: #{inspect(list)}. Sending acknowledgements...")
    messages
  end
First, in handle_message/3 we update our messages' data individually by converting them to
uppercase. Then, in the same callback, we're changing the success ack option of the message
to :term if its content is "FOO" or to :nack if the message is "BAR". In the end we
print each batch in handle_batch/4. It's not quite useful but should be enough for this
guide.

  
    
  
  Running the Broadway pipeline


Once we have our pipeline fully defined, we need to add it as a child in the supervision tree.
Most applications have a supervision tree defined at lib/my_app/application.ex.
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
You can now test the pipeline. Let's start the application:
iex -S mix

Use Gnat API to send messages to your stream:
Gnat.pub(:gnat, "test_subject", "foo")
Gnat.pub(:gnat, "test_subject", "bar")
Gnat.pub(:gnat, "test_subject", "baz")
Batcher should then print:
Got a batch: ["FOO", "BAR", "BAZ"]. Sending acknowledgements...


  

    
Gnat 
    



      
The primary interface for interacting with NATS

Gnat.ConnectionSupervisor 
    



      
A process that can supervise a named connection for you
If you would like to supervise a Gnat connection and have it automatically re-connect in case of failure you can use this module in your supervision tree.
It takes a map with the following data:
gnat_supervisor_settings = %{
  name: :gnat, # (required) the registered named you want to give the Gnat connection
  backoff_period: 4_000, # number of milliseconds to wait between consecutive reconnect attempts (default: 2_000)
  connection_settings: [
    %{host: '10.0.0.100', port: 4222},
    %{host: '10.0.0.101', port: 4222},
  ]
}
The connection settings can specify all of the same values that you pass to Gnat.start_link/1. Each time a connection is attempted we will use one of the provided connection settings to open the connection. This is a simplistic way of load balancing your connections across a cluster of nats nodes and allowing failover to other nodes in the cluster if one goes down.
To use this in your supervision tree add an entry like this:
import Supervisor.Spec
worker(Gnat.ConnectionSupervisor, [gnat_supervisor_settings, [name: :my_connection_supervisor]])
The second argument is used as GenServer options so you can give the supervisor a registered name as well if you like. Now in the rest of your code you can call things like:
:ok = Gnat.pub(:gnat, "subject", "message")
And it will use your supervised connection. If the connection is down when you call that function (or dies during that function) it will raise an error.

Gnat.ConsumerSupervisor 
    



      
A process that can supervise consumers for you
If you want to subscribe to a few topics and have that subscription last across restarts for you, then this worker can be of help.
It also spawns a supervised Task for each message it receives.
This way errors in message processing don't crash the consumers, but you will still get SASL reports that you can send to services like honeybadger.
To use this just add an entry to your supervision tree like this:
consumer_supervisor_settings = %{
  connection_name: :name_of_supervised_connection,
  module: MyApp.Server, # a module that implements the Gnat.Server behaviour
  subscription_topics: [
    %{topic: "rpc.MyApp.search", queue_group: "rpc.MyApp.search"},
    %{topic: "rpc.MyApp.create", queue_group: "rpc.MyApp.create"},
  ],
}
worker(Gnat.ConsumerSupervisor, [consumer_supervisor_settings, [name: :rpc_consumer]], shutdown: 30_000)
The second argument is a keyword list that gets used as the GenServer options so you can pass a name that you want to register for the consumer process if you like. The :consuming_function specifies which module and function to call when messages arrive. The function will be called with a single argument which is a Gnat.message/0 just like you get when you call Gnat.sub/4 directly.
You can have a single consumer that subscribes to multiple topics or multiple consumers that subscribe to different topics and call different consuming functions. It is recommended that your ConsumerSupervisors are present later in your supervision tree than your ConnectionSupervisor. That way during a shutdown the ConsumerSupervisor can attempt a graceful shutdown of the consumer before shutting down the connection.
If you want this consumer supervisor to host a NATS service, then you can specify a module that
implements the Gnat.Services.Server behavior. You'll need to specify the service_definition field in the consumer
supervisor settings and conforms to the Gnat.Services.Server.service_configuration type. Here is an example of configuring
the consumer supervisor to manage a service:
consumer_supervisor_settings = %{
  connection_name: :name_of_supervised_connection,
  module: MyApp.Service, # a module that implements the Gnat.Services.Server behaviour
  service_definition: %{
    name: "exampleservice",
    description: "This is an example service",
    version: "0.1.0",
    endpoints: [
      %{
        name: "add",
        group_name: "calc",
      },
      %{
        name: "sub",
        group_name: "calc"
      }
    ]
  }
}
worker(Gnat.ConsumerSupervisor, [consumer_supervisor_settings, [name: :myservice_consumer]], shutdown: 30_000)
It's also possible to pass a %{consuming_function: {YourModule, :your_function}} rather than a :module in your settings.
In that case no error handling or replying is taking care of for you, microservices cannot be used, and it will be up to your function to take whatever action you want with each message.

Gnat.Jetstream 
    



      
Provides functions for interacting with a NATS Jetstream
server.

Gnat.Jetstream.API.Consumer 
    



      
A module representing a NATS JetStream Consumer.
Learn more about consumers: https://docs.nats.io/nats-concepts/jetstream/consumers

  
    
  
  The Jetstream.API.Consumer struct


The struct's only mandatory field to set is the :stream_name. The rest will have
the NATS default values set.
Note that consumers are ephemeral by default. Set the :durable_name to make it durable.
Consumer struct fields explanation:
	:stream_name - name of a stream the consumer is pointing at.
	:domain - JetStream domain the stream is on.
	:ack_policy - how the messages should be acknowledged. It has the following options:	:explicit - the default policy. It means that each individual message must be acknowledged.
It is the only allowed option for pull consumers.
	:none - no need to ack messages, the server will assume ack on delivery.
	:all - only the last received message needs to be acked, all the previous messages received
are automatically acknowledged.


	:ack_wait - time in nanoseconds that server will wait for an ack for any individual. If an ack
 is not received in time, the message will be redelivered.
	:backoff - list of durations that represents a retry timescale for NAK'd messages or those being
 normally retried.
	:deliver_group - when set, will only deliver messages to subscriptions matching that group.
	:deliver_policy - specifies where in the stream it wants to start receiving messages. It has the
 following options:	:all - the default policy. The consumer will start receiving from the earliest available
message.
	:last - the consumer will start receiving messages with the last message added to the stream.
	:new - the consumer will only start receiving messages that were created after the customer
was created.
	:by_start_sequence - the consumer is required to specify :opt_start_seq, the sequence number
to start on. It will receive the closest available message moving forward in the sequence
should the message specified have been removed based on the stream limit policy.
	:by_start_time - the consumer will start with messages on or after this time. The consumer is
required to specify :opt_start_time, the time in the stream to start at.
	:last_per_subject - the consumer will start with the latest one for each filtered subject
currently  in the stream.


	:deliver_subject - the subject to deliver observed messages. Not allowed for pull subscriptions.
A delivery subject is required for queue subscribing as it configures a subject that all the queue
consumers should listen on.
	:description - a short description of the purpose of this customer.
	:durable_name - the name of the consumer, which the server will track, allowing resuming consumption
where left off. By default, a consumer is ephemeral. To make the consumer durable, set the name.
See naming.
	:filter_subject - when consuming from a stream with a wildcard subject, this allows you to select
a subset of the full wildcard subject to receive messages from.
	:flow_control - when set to true, an empty message with Status header 100 and a reply subject will
be sent. Consumers must reply to these messages to control the rate of message delivery.
	:headers_only - delivers only the headers of messages in the stream and not the bodies. Additionally
adds the Nats-Msg-Size header to indicate the size of the removed payload.
	:idle_heartbeat - if set, the server will regularly send a status message to the client while there
are  no new messages to send. This lets the client know that the JetStream service is still up and
running, even when there is no activity on the stream. The message status header will have a code of 100.
Unlike :flow_control, it will have no reply to address. It may have a description like
"Idle Heartbeat".
	:inactive_threshold - duration that instructs the server to clean up ephemeral consumers that are
inactive for that long.
	:max_ack_pending - it sets the maximum number of messages without an acknowledgement that can be
outstanding, once this limit is reached, message delivery will be suspended. It cannot be used with
:ack_none ack policy. This maximum number of pending acks applies for all the consumer's
subscriber processes. A value of -1 means there can be any number of pending acks (i.e. no flow
control).
	:max_batch - the largest batch property that may be specified when doing a pull on a Pull consumer.
	:max_deliver - the maximum number of times a specific message will be delivered. Applies to any
message that is re-sent due to ack policy.
	:max_expires - the maximum expires value that may be set when doing a pull on a Pull consumer.
	:max_waiting - the number of pulls that can be outstanding on a pull consumer, pulls received after
this is reached are ignored.
	:opt_start_seq - use with :deliver_policy set to :by_start_sequence. It represents the sequence
number to start consuming on.
	:opt_start_time - use with :deliver_policy set to :by_start_time. It represents the time to start
consuming at.
	:rate_limit_bps - used to throttle the delivery of messages to the consumer, in bits per second.
	:replay_policy - it applies when the :deliver_policy is set to :all, :by_start_sequence or
:by_start_time. It has the following options:	:instant - the default policy. The messages will be pushed to the client as fast as possible.
	:original - the messages in the stream will be pushed to the client at the same rate that they
were originally received.


	:sample_freq - Sets the percentage of acknowledgements that should be sampled for observability, 0-100.
This value is a binary and for example allows both 30 and 30% as valid values.

Gnat.Jetstream.API.KV 
    



      
API for interacting with the Key/Value store functionality in Nats Jetstream.
Learn about the Key/Value store: https://docs.nats.io/nats-concepts/jetstream/key-value-store

Gnat.Jetstream.API.KV.Watcher 
    



      
The watcher server establishes a subscription to the changes that occur to a given key-value bucket. The
consumer-supplied handler function will be sent an indicator as to whether the change is a delete or an add,
as well as the key being changed and the value (if it was added).
Ensure that you call stop with a watcher pid when you no longer need to be notified about key changes

Gnat.Jetstream.API.Object 
    



      
API for interacting with the JetStream Object Store
Learn more about Object Store: https://docs.nats.io/nats-concepts/jetstream/obj_store

Gnat.Jetstream.API.Stream 
    



      
A module representing a NATS JetStream Stream.
Learn more about Streams: https://docs.nats.io/nats-concepts/jetstream/streams

  
    
  
  The Jetstream.API.Stream struct


The struct's mandatory fields are :name and :subjects. The rest will have the NATS
default values set.
Stream struct fields explanation:
	:allow_direct - Allow higher performance, direct access to get individual messages. E.g. KeyValue
	:allow_rollup_hdrs - allows the use of the Nats-Rollup header to replace all contents of a stream,
or subject in a stream, with a single new message.
	:deny_delete - restricts the ability to delete messages from a stream via the API. Cannot be changed
once set to true.
	:deny_purge - restricts the ability to purge messages from a stream via the API. Cannot be change
once set to true.
	:description - a short description of the purpose of this stream.
	:discard - determines what happens when a Stream reaches its limits. It has the following options:	:old - the default option. Old messages are deleted.
	:new - refuses new messages.


	:discard_new_per_subject - - allows to enable discarding new messages per subject when limits are reached.
Requires discard: :new and the :max_msgs_per_subject to be configured.
	:domain - JetStream domain, mainly used for leaf nodes.
 See JetStream on Leaf Nodes.
	:duplicate_window - the window within which to track duplicate messages, expressed in nanoseconds.
	:max_age - maximum age of any message in the Stream, expressed in nanoseconds.
	:max_bytes - how many bytes the Stream may contain. Adheres to :discard, removing oldest or
refusing new messages if the Stream exceeds this size.
	:max_consumers - how many Consumers can be defined for a given Stream, -1 for unlimited.
	:max_msg_size - the largest message that will be accepted by the Stream.
	:max_msgs_per_subject - For wildcard streams ensure that for every unique subject this many messages are kept - a per subject retention limit.
Only available on nats-server versions greater than 2.3.0
	:max_msgs - how many messages may be in a Stream. Adheres to :discard, removing oldest or refusing
new messages if the Stream exceeds this number of messages
	:mirror - maintains a 1:1 mirror of another stream with name matching this property.  When a mirror
is configured subjects and sources must be empty.
	:mirror_direct - Allow higher performance and unified direct access for mirrors as well.
	:name - a name for the Stream.
See naming.
	:no_ack - disables acknowledging messages that are received by the Stream.
	:num_replicas - how many replicas to keep for each message.
	:placement - placement directives to consider when placing replicas of this stream, random placement
when unset. It has the following properties:	:cluster - the desired cluster name to place the stream.
	:tags - tags required on servers hosting this stream.


	:retention - how messages are retained in the Stream. Once this is exceeded, old messages are removed.
It has the following options:	:limits - the default policy.
	:interest
	:workqueue


	:sealed - sealed streams do not allow messages to be deleted via limits or API, sealed streams can not
be unsealed via configuration update. Can only be set on already created streams via the Update API.
	:sources - list of stream names to replicate into this stream.
	:storage - the type of storage backend. Available options:	:file
	:memory


	:compression - If file-based and a compression algorithm is specified, the stream data will be compressed on disk.
Valid options are "none" for no compression or "s2" for Snappy compression.
	:subjects - a list of subjects to consume, supports wildcards.
	:template_owner - when the Stream is managed by a Stream Template this identifies the template that
manages the Stream.

Gnat.Jetstream.PullConsumer behaviour
    



      
A behaviour which provides the NATS JetStream Pull Consumer functionalities.
When a Consumer is pull-based, it means that the messages will be delivered when the server
is asked for them.

  
    
  
  Example


Declare a module which uses Gnat.Jetstream.PullConsumer and implements init/1 and
handle_message/2 callbacks.
defmodule MyApp.PullConsumer do
  use Gnat.Jetstream.PullConsumer

  def start_link(arg) do
    Jetstream.PullConsumer.start_link(__MODULE__, arg)
  end

  @impl true
  def init(_arg) do
    {:ok, nil,
      connection_name: :gnat,
      stream_name: "TEST_STREAM",
      consumer_name: "TEST_CONSUMER"}
  end

  @impl true
  def handle_message(message, state) do
    # Do some processing with the message.
    {:ack, state}
  end
end
You can then place your Pull Consumer in a supervision tree. Remember that you need to have the
Gnat.ConnectionSupervisor set up.
defmodule MyApp.Application do
  use Application

  @impl true
  def start(_type, _args) do
    children = [
      # Create NATS connection
      {Gnat.ConnectionSupervisor, ...},
      # Start NATS Jetstream Pull Consumer
      MyApp.PullConsumer,
    ]
    opts = [strategy: :one_for_one]
    Supervisor.start_link(children, opts)
  end
end

  
    
  
  Connection Options


In order to establish consumer connection with NATS, you need to pass several connection options
via keyword list in third element of a tuple returned from init/1 callback.
Following options must be provided. Omitting this options will cause the process to raise
errors upon initialization:
	:connection_name - Gnat connection or Gnat.ConnectionSupervisor name/PID.
	:stream_name - name of an existing string the consumer will consume messages from.
	:consumer_name - name of an existing consumer pointing at the stream.

You can also pass the optional ones:
	:connection_retry_timeout - a duration in milliseconds after which the PullConsumer which
failed to establish NATS connection retries, defaults to 1000
	:connection_retries - a number of attempts the PullConsumer will make to establish the NATS
connection. When this value is exceeded, the pull consumer stops with the :timeout reason,
defaults to 10
	:inbox_prefix - allows the default _INBOX. prefix to be customized. Should end with a dot.
	:domain - use a JetStream domain, this is mostly used on leaf nodes.


  
    
  
  Dynamic Connection Options


It is possible that you have to determine some of the options dynamically depending on pull
consumer's init argument. To do so, it is recommended to derive these options values from some
init argument:
defmodule MyApp.PullConsumer do
  use Gnat.Jetstream.PullConsumer

  def start_link() do
    Gnat.Jetstream.PullConsumer.start_link(__MODULE__, %{counter: counter})
  end

  @impl true
  def init(%{counter: counter}) do
    {:ok, nil,
      connection_name: :gnat,
      stream_name: "TEST_STREAM_#{counter}",
      consumer_name: "TEST_CONSUMER_#{counter}"}
  end

  ...
end

  
    
  
  How to supervise


A PullConsumer is most commonly started under a supervision tree. When we invoke
use Gnat.Jetstream.PullConsumer, it automatically defines a child_spec/1 function that allows us
to start the pull consumer directly under a supervisor. To start a pull consumer under
a supervisor with an initial argument of :example, one may do:
children = [
  {MyPullConsumer, :example}
]
Supervisor.start_link(children, strategy: :one_for_all)
While one could also simply pass the MyPullConsumer as a child to the supervisor, such as:
children = [
  MyPullConsumer # Same as {MyPullConsumer, []}
]
Supervisor.start_link(children, strategy: :one_for_all)
A common approach is to use a keyword list, which allows setting init argument and server options,
for example:
def start_link(opts) do
  {initial_state, opts} = Keyword.pop(opts, :initial_state, nil)
  Gnat.Jetstream.PullConsumer.start_link(__MODULE__, initial_state, opts)
end
and then you can use MyPullConsumer, {MyPullConsumer, name: :my_consumer} or even
{MyPullConsumer, initial_state: :example, name: :my_consumer} as a child specification.
use Gnat.Jetstream.PullConsumer also accepts a list of options which configures the child
specification and therefore how it runs under a supervisor. The generated child_spec/1 can be
customized with the following options:
	:id - the child specification identifier, defaults to the current module
	:restart - when the child should be restarted, defaults to :permanent
	:shutdown - how to shut down the child, either immediately or by giving it time to shut down

For example:
use Gnat.Jetstream.PullConsumer, restart: :transient, shutdown: 10_000
See the "Child specification" section in the Supervisor module for more detailed information.
The @doc annotation immediately preceding use Jetstream.PullConsumer will be attached to
the generated child_spec/1 function.

  
    
  
  Name registration


A pull consumer is bound to the same name registration rules as GenServers.
Read more about it in the GenServer documentation.

Gnat.Server behaviour
    



      
A behavior for acting as a server for nats messages.
You can use this behavior in your own module and then use the Gnat.ConsumerSupervisor to listen for and respond to nats messages.

  
    
  
  Example


defmodule MyApp.RpcServer do
  use Gnat.Server

  def request(%{body: _body}) do
    {:reply, "hi"}
  end

  # defining an error handler is optional, the default one will just call Logger.error for you
  def error(%{gnat: gnat, reply_to: reply_to}, _error) do
    Gnat.pub(gnat, reply_to, "Something went wrong and I can't handle your request")
  end
end

      


      
        Summary


  
    Callbacks
  


    
      
        
  
    
    Gnat.Services.Server - gnat v1.10.0
    
    

    


  
  

    
Gnat.Services.Server behaviour
    



      
A behavior for acting as a NATS service
Creating a service with this behavior works almost exactly the same as Gnat.Server,
with the bonus that this service keeps track of requests, errors, processing time, and
participates in service discovery and monitoring as defined by
the NATS service protocol.

  
    
  
  Example


defmodule MyApp.Service do
  use Gnat.Services.Server

  # Classic subject matching
  def request(%{body: _body, topic: "myservice.req"}, _, _) do
    {:reply, "handled request"}
  end

  # Can also match on endpoint or group
  def request(msg, "add", "calculator") do
    {:reply, "42"}
  end

  # defining an error handler is optional, the default one will just call Logger.error for you
  def error(%{gnat: gnat, reply_to: reply_to}, _error) do
    Gnat.pub(gnat, reply_to, "Something went wrong and I can't handle your request")
  end
end

